

# Macro: lfit\_simple.mac

Macro	lfit_simple.mac
Description	Demonstrates fitting a line to pass between two given points
CM version	Any

## What the macro does

This macro attempts to fit a straight line between two given points. Figure 1 shows three stages in the process. On the left is the initial state showing the two points and the line. On the right is the final result with the line having moved between the points.

Two constraint rules are applied. One says that one end of the line has to lie at one of the points; the other says that the other end must lie at the other point. The constraint modeller tries to resolve these constraints by adjusting the position and rotation of the line. The middle part of figure 1 shows various trial positions as the modeller attempt the resolution.




Figure 1: Graphics screen before, during and after fitting the line

## How the main part of the macro works

The listing of the macro is given below. The lines of the macro are numbered for ease of reference. As is often the case, much of the macro consists of commands to make the application look good on the screen.

At the beginning of the macro (lines 0009–0010) a number of global variables are declared. Of these, `mm` is a “model space” and this is effectively a transformation. The two points are `p1` and `p2` and the line is `l1`.

A user-defined function (UDF) called `setup` is defined in lines 0012–0022. Line 0014 defines the model space. This is a two-dimensional model space. Entities can be embedded into such a space. The space has a transformation associated with it. This combines translation in the  $x$ - and  $y$ -directions, and a rotation about the  $z$ -axis (the axis perpendicular to the plane). It is also possible to define a global scale factor for the model space. Initially the scale factor is unity. The command at line 0014 defined the model space `mm` with zero translation in  $x$  and  $y$  (the first two arguments of the `mod2` function), and zero angle of rotation (the third argument). The two points are defined in lines 0015 and 0016. Then the line `l1` is defined. The arguments of to the `lin` function are the  $(x, y, z)$ -coordinates of its two end-points. These are chosen slightly strangely, but this is so that the line initially appears away from the two points. The line passes between the points  $(-8, -3, 0)$  and  $(-3, -3, 0)$ ; thus it has length 5 units. The final argument to the `lin` is the model space `mm`. This means that the line is embedded in this model space and when the model space is changed, the line moves. Lines 0018–0021 of the macro change the “font” of the points so that they appear as dots on the screen, and the colours of these points and of the line.

Next in the macro file comes the function `linefit` (lines 0025–0034). This creates two constraint rules (lines 0031 and 0032). These use the function `on` which is a built-in function of the modeller. It finds the distance (in world space) between two geometric objects. The first rule gives the expression which is the distance between the first end-point of line `l1` (denoted by `l1:e1`) and the point `p1`. If this distance is zero, then the end of the line lies at the point. Similarly the second rule expression is for the distance between `l1:e2`, the second end-point of the line, and the point `p2`.

The modeller needs to know what it can change when trying to resolve the constraints. This is done with the `var` list (line 0027). This only gives the model space `mm`. There are several components to this model space, but, by default, the only ones which are free are the two components of translation and the angle of rotation.

The function `linefit` takes one argument when it is called. This is specified by the `inp` statement (line 0029) which says that there is one argument which is called `code` within the function and is an integer as specified in line 0028. This variable is used with the `rpnt` function in line 0033. If `code` is 1, then `rpnt` clears the screen and repaints it; if it is zero, then the screen is repainted without clearing the screen; and if it is  $-1$ , then the `rpnt` function does nothing.

The `linefit` function could be invoked by entering the following command in the command window.

`linefit( 0 )`

However, to help the user, a menu is created.

## How the rest of the macro works

Lines 0037–0081 create the menu with its buttons and submenus.

The submenu **Show working** (lines 0047–0054) calls `linefit` with the argument set to zero and to unity. The former allows all the attempts at repositioning the line to be shown (as in the middle of figure 1, but this tends to slow down the resolution process. The latter again shows all the attempts but as an animation.

Lines 0055–000072 enable the user to change the positions of the two points. Here the built-in function `dodig` is used. This gives a prompt to the user and then waiting for the user to move the cursor and click one of the mouse buttons. The function `getdig` retrieves the position of the cursor and uses this to redefine the appropriate point. As the point has been redefined, its colour and font have to be redefined.

Initially the modeller succeeds in fitting the line between the original points. This is because the points are 5 units apart which is also the length of the line. Once the points are moved, the modeller places the line midway between the points. The model space `mm` also has a component called `mm:sigma` which represents a (global) scale factor. Allowing this to vary enables the line to be stretched or shrunk. Lines 0073–0081 allow the user to free or fix this scale factor.

In line 0084, the `remmenu` command is used to remove any existing user menu, and then, in line 0085, the `addmenu` command is used to show the newly defined menu.

After this, the `setup` function is called to define the initial positions of the geometric objects. The the graphics window is created and the screen repainted. The `zoom` command repaints the graphics at full size for the screen, and the second `zoom` command shrinks it to a somewhat smaller size.

GM

May 2013

## Listing

```
0001 $ =====
0002 $   LFIT_SIMPLE.MAC
0003 $ =====
0004 $   Fitting a line between two points
0005 $   October 1997
0006 $   Revised: May 2013
0007 $ =====
0008
0009 dec mod2    mm;                $ global variables
0010 dec geom    p1, p2, ll;
0011
0012 function setup                  $ start of function
0013 {
0014     mm = mod2( 0, 0, 0 );        $ define model space
0015     p1 = pnt( 4, 0, 0 );        $ define first point
0016     p2 = pnt( 0, 3, 0 );        $ define second point
0017     ll = lin( -8, -8, 0, -3, -8, 0, mm ); $ define the line
0018     cfont( 6, p1, p2 );        $ points are dots
0019     ccol( green(), ll );        $ line is green
0020     ccol( red(), p1 );          $ first point is red
0021     ccol( yellow(), p2 );       $ second point is yellow
0022 }                                $ end of function
0023
0024
0025 function linefit                $ start of function
0026 {
0027     var mm;                     $ var list: just mm
0028     dec int code;               $ declare local integer
0029     inp code;                   $ use it as argument
0030
0031     rule( ll:e1 on p1 );        $ first constraint rule
0032     rule( ll:e2 on p2 );        $ second constraint rule
0033     rpnt( code );               $ repaint the screen
0034 }                                $ end of function
0035
0036
0037 menu lfit_menu                  $ start new menu
0038 {
0039     button Fit the line         $ define button
0040     { linefit(-1);              $ call linefit function
0041       rpnt(1);                  $ clear and repaint screen
0042     }
0043     button Reset                $ define button
0044     { setup();                  $ call setup function
0045       rpnt(1);                  $ clear and repaint screen
0046     }
0047     submenu Show working>      $ define submenu
0048     { button Overlay            $ define button
0049       { linefit(0);             $ call linefit function
0050     }

```

Figure 2: Listing of macro lfit\_simple.mac (part 1)

```

0051         button Animate                                $ define button
0052         { linefit(1);                                $ call linefit fucntion
0053         }
0054     }
0055     submenu Move points>                                $ define submenu
0056     { button Red point                                $ define button
0057         { if( dodig( "Give new red point" ) )          $ go into digitize mode
0058             { p1 = getdig();                            $ get digitize, redefine p1
0059                 ccol( red(), p1 );                      $ reset colour of p1
0060                 cfont( 6, p1 );                          $ and its font
0061                 rpnt(0);                                $ repaint without clearing
0062             }
0063         }
0064         button Yellow point                            $ define button
0065         { if( dodig( "Give new yellow point" ) )        $ go into digitize mode
0066             { p2 = getdig();                            $ get digitize, redefine p2
0067                 ccol( yellow(), p2 );                    $ reset colour of p2
0068                 cfont( 6, p2 );                          $ and its font
0069                 rpnt(0);                                $ repaint without clearing
0070             }
0071         }
0072     }
0073     button Free scale                                $ define button
0074     { free( mm:sigma );                                $ free global scale in mm
0075         fwriteln( 0, "Scale freed" );                    $ output message to screen
0076     }
0077     button Fix scale                                $ define button
0078     { fix( mm:sigma );                                $ fix global scale in mm
0079         fwriteln( 0, "Scale fixed" );                    $ output message to screen
0080     }
0081 }
0082
0083 remmenu();                                            $ remove any previous menu
0084 addmenu( lfit_menu );                                $ add new menu
0085
0086 setup();                                            $ call setup function
0087
0088 graphics();                                            $ create graphics window
0089 rpnt(1);                                            $ clear and repaint
0090 zoom();                                            $ zoom all
0091 zoom( 0.75 );                                        $ and zoom down a little
0092
0093 $ End of file
0094

```

Figure 3: Listing of macro lfit\_simple.mac (part 2)